
INTRODUCTION
Welcome to the world of BC BASIC programming. This package will
enable you to get the most from your Commodore 64, whether you

are a computing expert or just a beginner. BC BASIC maken the
64°s advanced facilities, such as hi-res graphics, sprites anl
sound, easy and simple to use by means of many fast and powerful
commands and functions added to the BASIC language. Whatever you
use your 64 for, from writing games to serious applications, W
BASIC will help you do it much more easily and quickly.

This manual is arranged so that it begins with the almplo
features of BC BASIC, and then progresses to more complex onon

To get the most from BC BASIC, work through the manual, trylny
out the examples and programs of your own so that you can wee
for yourself how each command or function works. At the end ol
each chapter, you may like to try to write some programs umlng
what you have learnt so far. You may be quite surprised by |l
standard of programs you will be writing after a short whilol
Even advanced programmers are advised to read through {he
chapters in sequence, as later chapters will refer to commanis
and functions explained earlier on.

Now it just remains for me to wish you every success with your W
BASIC programming.

Brian Candler,
author of BC BAHIC,

INSERTING THE CARTRIDGE

1. TURN THE MACHINE OFF.

2. Insert the cartridge into the cartridge port - this is the
large rectangular hole on the back of the 64. Press it home
firmly.

3. Now turn the 64 on. After a couple of seconds, the BC BASIC
startup message will appear. BC BASIC is now ready for use.

REMEMBER -NEVER INSERT OR REMOVE THE CARTRIDGE WITH THE MACHINE
SWITCHED ON, OR YOU WILL RISK DAMAGING THE CARTRIDGE,
YOUR COMMODORE 64, OR BOTH.

Chapter 1

BASIC GRAPHICS

Now you have got BC BASIC up and running, here are a few simple
but useful commands to get you started. Try them out at the
keyboard.

BORDER col This changes the border colour of the screen
to ‘col”, where col is in the range 0-15 and
represents one of the following colours:

0 Black 8 Orange

1 White 9 Brown

2 Red 10 Light red

31 Cyan 11 Dark grey

4 Purple 12 Medium grey
5 Green 13 Light green
6 Blue 14 Light blue
7 Yellow 15 Light grey

‘col” does not have to be a number - it may
be a variable or expression, like B or C*2+3.
However, if it is outside the range 0-15 an
?ILLEGAL QUANTITY error will result.

Whenever you see ‘col” again in a BC BASIC
command definition, you know it will mean a
number or expression in the range 0-15.

Try typing BORDER 13 - this will change the
border of the screen from its normal medium-
grey colour to light green. Pressing
RUN/STOP and RESTORE together will reset the
screen colours to a dark grey (11)
background, with a medium grey (12) border
and cyan (3) letters.

PAPER col This command, similar to BORDER above,
changes the screen background colour (the
‘paper” on which the computer writes) to col.

BORDER and PAPER are the same as POKE 53280,
col and POKE 53281,col respectively, but much
easier to remember!

INK col This command sets what colour is to be
printed to the screen. For example, typing
INK 14 makes "READY." and all further
characters appear as light blue, rather than
cyan.




Try the following program, which displays all
of the colours available on the 64:

10 FOR C=0 TO 15

20 INK C
30 PRINT "COMMODORE 64!"
40 NEXT C
AT y,X This command instantly moves the cursor to
the xth position along line y. It maken
moving characters and positioning titles otc,
easy. y must be in the range 0 to 24, and x

must be in the range 0 to 39. 0,0 is the top
left corner of the screen. y comes before x
for two reasons:
a). For convenience (in being able to
specify the line before the column)
b). For compatibility with BASICs on
other machines.

AT may be used on its own (ready for
PRINTing) or actually inside a print
statement:

e.g. PRINT AT 3,0;"HI";AT 5,4;"THERE!"

THE SAME APPLIES TO BORDER, PAPER AND INK.

INK is quite commonly used inside a PRINT
statement.

Short forms for keywords

You may have met abbreviations for commands and functions in
standard Commodore BASIC - e.g. ? for PRINT and G shift-O for
GOTO. Many BC BASIC commands can be abbreviated in a shortened
form. Of those you have met so far:

BORDER may be abbreviated to B shift-0
PAPER may be abbreviated to P shift-A

From now on, if a command or function has a short form, it will
be printed below the definition in brackets, like this:

BORDER col
(bO)

where a capital O indicates that it must be shifted.

Animation using PRINT AT |

This is far easier than anything you may have tried using POKE.
Just follow these steps:

1. Put a character on the screen at a certain position (using
AT) X

24 Wait for a moment.

3. Print a space at the same place as the character (this wipes
it off)

4. Change its position and go back to step 1.
For example, to move a star across the screen:

10 Y=5:X=0:REM FIRST CHARACTER ON LINE 5

20 PRINT AT Y,X;"*"

30 FOR K=1 TO 50:NEXT K:REM A SHORT DELAY {
40 PRINT AT Y,X;" "

50 X=X+1:REM MOVE IT ACROSS

60 IF X<40 THEN GOTO 20

This program prints a star at 5,0, waits a moment, wipes it off,
and then adds 1 to X (so it will next be displayed at 5,1 instead
of 5,0) and carries on until it reaches the right-hand side of
the screen, when it stops.

Some improvements

5 PRINT "[cls]" will clear the screen at the beginning of the
program. [cls] appears as a reverse-heart
symbol, and is obtained by pressing SHIFT and
CLR/HOME together.

Changing the number ‘50" in line 30 will alter the speed of
movement - the smaller the number, the less the delay, and so the |
faster the star will move. If you make the delay too small,
however, the character will be wiped off the screen almost as

soon as it was put there, and so the star will flicker
appreciably.

Try changing the character printed in line 20. You can print
something longer than 1 character (as long as there are the same
number of spaces in line 40) so you could even have your name
moving across the screen!

of course, as well as being able to put characters on the screen, i
it would be nice to be able to determine what character is at a
certain position on the screen. The following definitions are

for functions, as opposed to commands - they may only be used

within expressions (e.g. LET A=GCOL(L,C) ) |



SCR(y,x) This function returns the "POKE code” of the
character at a certain screen location y,x: thins
is the code you would POKE to screen memory to
make that character appear. There is a table of
POKE codes (or ’‘screen display codes’) in the
Commodore 64 User Manual - however, the function
CODE, described below, will find them for vyou.
Like AT, y and x are numeric expressions where y
is in the range 0-24 and x in the range 0-39.

An example using SCR will follow shortly.

GCOL(y,x) This function, which has the same format as SCR,
(gC) returns the colour of the character at y,x, as a
number 0-15. GCOL stands for Graphics COLour.

e.qg. if PRINT GCOL (3,5) prints 14, then the
character on line 3, column 5 is light-blue.

CODE(n) This function 1is used to convert characters to
or CODE(n$) their POKE codes. If there is a number or numeric
(coD) expression in the brackets, this is taken as an
ASCII code and converted to its associated POKE
code. If n is not a printable ASCII character,
i.e. if it is out of the range 0-255 or is a
control character, then an ?ILLEGAL QUANTITY error
is printed.
The second format shown, with a string inside the
brackets, converts the first character of the
string to its POKE code, unless the first
character is a [rvs] or [rvs-off] character, in
which case the second character is converted, and
128 added to the code if the first character was
[rvs]. Again, if the character does not have a
POKE code, an ?ILLEGAL QUANTITY error will result.

A few examples should clear things up:
PRINT CODE(65) prints 1, since 65 is the ASCII e
code for an ‘A", and 1 is the
POKE code for this character.
A=CODE("2") Sets A to 26 1
A=CODE("[rvs]Z") Sets A to 154 (=26+128)
PRINT CODE(17) Gives an error, since 17 is a ol
control code, not a printable
character.
_ = === ll |
An example of the use of SCR - collision detection r
If you use SCR just before PRINT AT, you can find out what
character you are printing on top of. We can now modify the -
‘moving star’ program so that it crashes against a mountain.
‘ =
6

10 PRINT "[cls]”

20 PRINT AT 7,0;" XXX"

30 PRINT AT 8,0;" XXX XXXXXXX XXX"

40 PRINT AT 9,0;" XXXXXXXXXXXXXXXXXXXXXXXXXXXXXX"
50 P=0:REM P IS POSITION ALONG LINE

60 IF SCR(7,P)<>CODE(" ") THEN GOTO 120:REM CRASH?
70 PRINT AT 7,P;"*"

80 FOR K=1 TO 50:NEXT K

90 PRINT AT 7,P;" "

100 P=P+1:REM MOVE IT ALONG

110 GOTO 60

120 PRINT AT 7,P-2;"CRASH!"

130 AT 20,0:END

Lines 10-40 of this program clear the screen and display a
terrain. Replace the Xs with graphic symbols - chequered flags
perhaps (press the Commodore and “+° keys together).

Line 50 sets the initial position along the line.

Line 60 is the important one. It says "if the character where I'm
just about to put a star is not a space, then goto line 120" - in
other words, when the star reaches the terrain, it stops moving
and the program continues on line 120.

Lines 70-110 are the same as the old moving star program, except
without the test to see if we are at the edge of the screen (it
is unnecessary, since we are going to crash first). |
Line 120 prints CRASH! where the star was, and line 130 moves the
cursor to the bottom of the screen before stopping.

In the next chapter, we shall find out how to overcome the
limitations of only having the graphics available on the
keyboard, by being able to define your own shapes.

Exercise - try bouncing a character backwards and forwards (hint
- split the program into two parts, one to move left to right and
the other to move back again).




Chapter 2

USER-DEFINED GRAPHICS

These are a powerful feature of the 64 which allow you to create
your own characters and symbols. Firstly, however, we mus
digress a little.

Binary and hexadecimal notation

BINARY, or Base 2, is the system of numbering which the computer
uses. It is ideal for this purpose, since it only consists of Os
and 1ls, and can be stored in the computer as ‘Off " or ‘On’.

1 binary digit = 1 “bit”

1 byte = 8 bits. Each memory location in the computer is |1

byte wide. 1K, or kilobyte, equals 1024 bytes.

BC BASIC uses the % function to convert binary to decimal. So:
PRINT %1101

prints 13 (13=8+4+1).

HEXADECIMAL, or just HEX for short, is a convenient way of

representing computer numbers without strings of Os and 1s. L

is actually base 16 - each digit may be 0-9 or A-F. Hex numbers

are preceeded by the °“$° sign.

e.g. PRINT $4D2

prints 1234 (1234=4*256+13*16+2)

Don’t worry if this all seems confusing - it is not wvital to

understand. However, you should know that memory addresses range
from $0000 to $FFFF - a total of 65536 bytes.

Here is a ‘memory map - a chart of how the memory is arranged -
on a Commodore 64 under BC BASIC, as the 6510 processor chip sees
5 = Again, do not worry if you do not understand it all -

however, note the position of USER RAM as this is important.

4K INPUT/OUTPUT (optionally 4K CHARACTER :

5 GENERATOR ROM)

: $0400-$07FF : Normal position of SCREEN RAM and Sprite :
2 : pointers i

This, on the other hand, is how the VIC chip sees the 64°'s
memory :

BLOCK 2 $A000-$BFFF RAM :

$9800-$9FFF CHARACTER ROM e

(Lower case). :

$9000-$97FF CHARACTER ROM 5

(Upper case). :
$8000-$8FFF RAM.

$8000-$BFFF

$2000-$3FFF RAM.
$1800-$1FFF CHARACTER ROM

$0000~-$3FFF 4
(Lower case) ;

BLOCK 0

$1000-$17FF CHARACTER ROM
(Upper case)
$0000-$OFFF RAM.

w
et
o
Q
=
—
bl
>
=

There are several very important points to notice on this
diagram.

Firstly, the VIC chip cannot ‘see’ the BASIC and KERNAL ROMs - it
just sees the RAM which sits behind them. This memory is used by
BC BASIC for hi-res information, so as not to waste BASIC memory.

Secondly, notice that the diagram is divided into four 16K
BLOCKS . THE VIC CHIP CAN ONLY ACCESS ONE OF THESE BLOCKS AT A




TIME. Therefore, all graphics, screen data and sprites must be
in the same 16K block - if you wish to move one of them out into
a different block, you must move all the rest at the same time
too.

Thirdly, the VIC chip cannot see RAM at $1000-$1FFF or §9000
$9FFF - it sees the character generator ROM instead. Therefore,
you cannot put your user-defined graphics data there (you would
be overwriting your BASIC program anyway!) so the user-definad
graphics commands will return an error if you try to; you will
have to move the character set first. All will be explained
shortly.

User-defined graphics - step-by-step

Basically, this is an all-or-nothing process on the Commodore 64.
You must tell the VIC chip to get its character data from a
different area of RAM, copy the ROM characters there and then
change the ones you want.

a). Where do I put the character set?
To begin with, it would obviously be simplest if we were to keep
the VIC chip in the same 16K block that it is used to - block 0,
which is at $0000-$3FFF (see diagram above). A whole character
set takes 2K, or $800, bytes. Therefore, we shall put the
character set at $3800-$3FFF and tell BASIC not to use any memory
above $37FF. This obviously wastes memory, but it is simplest to
begin with, and the 12K which is left is sufficient for programs
of a reasonable length. Later in this chapter we shall see how
to put the character set right at the top of user RAM.

b). What do I do next?

Firstly, we must tell BASIC that the RAM from $3800 upwards is
reserved, and not to be interfered with. This is done by the
following command:

HIMEM=$3800
(hI)

Please note that this command clears all variables - therefore it
should be the first line of a program, or should be typed in
immediate mode (directly at the keyboard), when the contents of
variables do not matter. Try it now.

Now we must tell VIC where to get its characters from - this is
done using the CHARSET command. To find what number should
follow CHARSET, take the address of the character set and divide
it by 2048 ($800) - in our case, this is $3800/$800, which turns
out to be 7 (try it for yourself).

N.B. Your character set MUST start on a 2K boundary - i.e. it
must start at $0000,$0800, $1000, $1800, $2000 etc.

Now you can type the following command to move the character sot:

10

CHARSET 7

(chA)
Oh dear! The screen seems to have turned into a dreadful mess.
Do not worry - we haven’'t copied the ROM characters into RAM yet,

remember?
Typing the following remedies the situation:

COPY 0 TO 255
(coP)

It will now seem that you are in normal mode with only 12K of
memory free - but you are now ready for the next stage.

HINT: To copy your characters from the 1lower case (small
letters) character set instead of the upper case one, use
instead:

CoPY £1,0 TO 255 [COPY £0 may be used instead of

COPY - i.e. COPY £0,0 TO 255 =
COPY 0 TO 255]

e Defining your own characters

This is the fun part! Firstly, draw out your character as an 8
by 8 square matrix, as follows:

A coloured-in block represents a dot on the screen °‘ON° or
illuminated in the character colour - a blank square represents a
screen dot in the background or PAPER colour. You may find
graph-paper useful for this operation. You are recommended to
make all vertical lines in your character at least 2 dots wide to
help prevent ’‘chroma noise” (colour distrotion) on your TV.
This is not necessary if you are using a black-and-white TV.

Now decide which character you are going to substitute the ’Space
Invader ~ or whatever for - let’s say the “*°. You must first
find the POKE code of the character, using the CODE function, as
described in the previous chapter. Typing:

PRINT CODE("*")

11




will give the required number - 42 in this case. Now vyou can
define your character in rows, using ‘0" for a space and "1 ° for
a dot on the screen:

DEFUSR 42;0;%00011000
DEFUSR 42;1;%00111100
DEFUSR 42;2;%01111110
DEFUSR 42;3;%11011011
DEFUSR 42;4;%11111111
DEFUSR 42;5;%11000011
DEFUSR 42;6;%01111110
DEFUSR 42;7;%11000011

(dEuS)
From now on, every time you type a “*°, a space-invader, or
whatever character you defined, will appear. The number
sandwiched between semicolons (and they must be semicolons) is

the row of the character being defined. The % indicates binary,
and converts the binary digits following it to decimal. Each row
is 8 bits, or 1 byte.

The change is actually an illusion - using a “*° in BASIC
programs will work as normal, although it will look odd. If you
want to copy the normal “*° back from ROM, to erase your
character, use the COPY command:

COPY 42

To return to normal ROM characters, type CHARSET 2

To switch to lower-case ROM characters, type CHARSET 3

(this is because 2*$800=$1000 and 3*$800=$1800, and if you refer
back to the 64 ‘s memory map as VIC sees it, this is the position
of the ROM character set).

After this, to reset the top of memory pointer (so BASIC can use
all its RAM again) type:

HIMEM=$8000
The DEFUSR command is more flexible than shown above
- it can take more than one row of data at a time (separate the

rows by commas) and, if no row number is specified, 0 is assumed.
So:

DEFUSR 42,%00011000,%00111100,%01111110,%11011011
DEFUSR 42;4;%11111111,%11000011,%01111110,%11000011

has the same effect as the 8-line character definition for the
space-invader given previously. Note that you cannot go past row
7 when using DEFUSR.

Some examples:

DEFUSR 81,56,120,199,192,199,120,56,0

12

defines code 81 (a shift-Q) as a Commodore logo.

10 FOR R=0 TO 7
20 READ A$:A=VAL("3"+AS$)

30 DEF USR 131;R;A {
40 NEXT R

50 DATA 00111100,01000010,10011001,10100001

60 DATA 10100001,10011001,01000010,00111100

This defines code 131 (reverse-C) as a Copyright symbol. Notice
in 1line 20 that you can use % inside VAL, to convert binary
strings to decimal (the same applies to $) and in line 30 that
DEFUSR can be split into two words.

Reading the character set

It 1is useful to be able to read the definitions of characters - {
this is performed by the following function.

CHARSET(ch,rw) This function fetches row ‘rw’ of character
(cha) ‘ch® in the current character set, and
returns it as a value in the range 0-255,
where each bit of the number corresponds to a
dot in the character. ‘ch” must be between 0
and 255, and ‘rw’ must be between 0 and 7.
‘ch” must be the POKE code (obtained by the {
CODE function) of the required character.

The next function is wuseful in the context of reading and
displaying characters, as it converts decimal to binary.
NOTE: IN BC BASIC command and function definitions, anything in !

square brackets [ ] is ogtlonal - it may be left out if desired.
BIN$ (n[,d[,x,y]]) This function returns the number ‘n”°, which
(bI) may be between -$FFFFFFFF and +$FFFFFFFF, as

a binary string. If n is positive, the first
character of the string is a space; if n is
negative, it is a minus sign (just 1like
STR$). If d is specified (in the range 0-32)
it is the minimum number of digits which will {
be returned - padded with zeros if necessary.
If x and y are specified (each must be
between 0 and 255) they are the ASCII codes
which will be returned instead of the 0 and 1
characters respectively - i.e. you can
substitute characters of your own choice for
the 0 and 1 symbols.

e.g. PRINT BINS(5) prints 101
A$=BINS$(53,8) sets A$ to " 00110101"
PRINT BIN$(53,12,46,42)

Prints . .ues NAL B W
(since 46 is the code for “.  and 42 is
the code for "*7).

13




Here is an example program using CHARSET and BIN$. It prints the
64 ‘s character set continuously as large (8 by 8) characters.

10 PRINT"[cls]"

20 FOR C=0 TO 255:REM DISPLAY ALL CHARACTERS
30 AT 0,0:REM MOVE CURSOR TO TOP OF SCREEN
40 FOR R=0 TO 7:REM R IS A ROW COUNTER

50 PRINT BINS$(CHARSET(C,R),8,46,209)

60 NEXT R:REM PRINT NEXT ROW

70 NEXT C:REM NEXT CHARACTER

80 GOTO 20

Line 50 is the vital line in this program. It says:
"Print in binary the Rth row of character C in the character set,
as an 8-digit binary number, printing character code 46 (a full

stop) for a 0 and character code 209 (a shift-Q symbol) for a 1"

Moving VIC into a different 16K block

If you try to use CHARSET followed by a number greater than 7,

you will get an ?ILLEGAL QUANTITY error - this is because
8*$800=$4000 - i.e. you have tried to move the character set into
the next 16K block. In fact, this is possible, but only if you

move the screen RAM (the area where the characters on the screen
are kept) IN THE SAME CHARSET COMMAND, TO THE SAME 16K BLOCK AS
THE CHARACTER SET. To do this, you will need a full definition
of the CHARSET command :

CHARSET[cs ][ ,sr] This command is used to move the position of
(chA) the character set, and/or the position of the
screen RAM.
cs, between 0 and 31, is used to specify the

new position of the character set. This is
found by dividing the address of the
character set by $800. e.g. to put the
character set at $7800, cs would be

$7800/$800, which is 15.
sr, between 0 and 63, is used to specify the
new position of the screen RAM. It is found
by dividing the address of the screen RAM by
$400.
NOTES
Although both parameters c¢s and sr are
indicated ‘as being optional, at least one
must be present. If one is omitted, the
current value of the other one is taken.
THE CHARSET COMMAND MUST RESULT IN BOTH THE
CHARACTER SET AND SCREEN RAM BEING IN THE
SAME 16K BLOCK, or an error will result.
So CHARSET 15,29 is fine, since:
15*$800=$7800
and 29*$400=$7400, and these are in the
same 16K block.

But CHARSET 15,1 or CHARSET 15 while the
screen RAM is not in the $4000-$7FFF block
will both return errors.

(HINT: When dealing with user-defined graphics, or UDGs, you may
find the ‘@  function useful - this converts decimal numbers to
hex. So PRINT @1234 prints 4D2; PRINT @(15*$800) prints 7800.)

NOW: to waste as little memory as possible, the screen RAM and
character set should go to the top of user RAM. We shall put the
screen RAM at $7400 and the character set at $7800-$7FFF.

HIMEM=$7400:CHARSET 15,29:COPY 0 TO 255:PRINT"[cls]"

The HIMEM command is to protect the RAM against interference from
BASIC; the CHARSET command moves the character set and screen
RAM, as explained above; the COPY command copies the ROM
characters into RAM; and then the screen is cleared to remove any
initial garbage. This line is the standard first line in any UDG
program.

The default (normal) CHARSET setting is CHARSET 2,1 (i.e. the
character set is at $1000, where the VIC chip sees the character
ROM, and the screen RAM is at 1*$400=$0400). So, if you wish to
revert to normal characters at the end of a program, use:

CHARSET 2,1:HIMEM=$8000:PRINT" [cls]":END

All these functions, apart from the HIMEM command, are performed
by pressing RUN/STOP and RESTORE together.

Now you have a set of powerful user-defined graphic commands at
your disposal; chapter 12 deals with more advanced character
graphics, such as defining your own multi-colour characters.

15

Chapter 3

BASIC SPRITES

Movement using PRINT AT and user-defined characters is all very
well, but it has its drawbacks. Firstly, movement tends to be
jerky, as characters must be moved from one screen location to

the next - i.e. 8 dots at a time. UDGs are small - only 8 by 8
dots, and making larger shapes by joining up smaller ones is
tedious. Also, moving characters with PRINT AT tends to wipe

away whatever was behind them (e.g. if you start with a starry
background, the stars will gradually disappear as characters move
over them). i

As a solution to all these problems, the VIC chip enables you
to use Sprites (or MOBs, Moveable Object Blocks, as Commodore
prefers to call them). These are a powerful graphic feature
practically unique to the Commodore 64. You can display up to 8
sprites at one time; each is defined on a large, 21 by 24 matrix
of dots; they may be positioned anywhere on the screen, not being
limited to 8 by 8 character squares and not interfering with
screen data; they may be optionally expanded x2 and in the x-
and/or y-directions; they have priorities, i.e. some go in front
of others; they may be multi-coloured; and there is automatic
collision-detection between sprites and the background and
sprites and other sprites. Sprites offer a quick and easy way to
create superb smooth movement and animation.

This chapter will deal with the basic Sprite facilities;
advanced sprites, including interrupt-driven (automatically
moving) sprites, will be dealt with in chapter 9.

a). Where do I put my sprites in memory?

A sprite definition is 64 bytes (actually 63 bytes plus 1 spare),
and there are 256 such ‘sprite blocks’, numbered 0-255, in each
16K block of memory which the VIC chip can access. Fortunately,
you do not have to reserve memory for sprites (although you can
if you want to) because there is some memory free. Sprite block
11 is free, whatever happens. Sprite blocks 13 to 15 inclusive

are also free, but these are in the cassette buffer, and are
corrupted by tape operations such as LOAD and SAVE. Similarly to
user-defined graphics, sprites must be defined on a 64-byte
boundary, i.e. at $0000,$0040,$0080 etc.

If you have moved VIC away from block 0, either by using CHARSET
or because you are in a hi-res graphics mode, you will have to
find other memory space for your sprites. This will be explained
in chapter 9.

b). How do I put them there?

The process is similar to defining user-defined graphics, except
you use the DEFSPR command.

Firstly, draw up a 21 by 24 matrix of squares, like that on page
70 of the Commodore 64 User Manual, and define your sprite on it.
(N.B. The rows are numbered 0-20 in BC BASIC, not 1-21 as in
the manual).

Now you can define your sprite:

DEFSPR 13,0,%000000000111111100000000
DEFSPR 13,1,%000000011111111111000000
DEFSPR 13,2,%000000111111111111100000
etic.

This defines the first 3 rows of the ‘Commodore balloon” in the
manual, into sprite block 13. Carry on until you have done row
20 - it is a long job, but worth it.

If you want to keep your sprite, put all these DEFSPR commands as
program lines, and then run the program. You can then save the
program later, or edit the sprite.

HINT: To clear a sprite, use the following line:
FOR R=0 TO 20:DEFSPR 13,R,0:NEXT R

In this DEFSPR command, 13 is the number of the block being
cleared, R 1is a row counter variable, and 0 is the same as
%000000000000000000000000.

A more efficient way of defining sprites will be dealt with later
in this chapter.

c). Now how do I display it on the screen?

There are 8 sprites, numbered 0 to 7. These may all be used, but
for the moment we will just use one of them. Choose one - let’s
say 0 for simplicity.

Firstly, you must switch it on - this is done by the SPRON
command (all sprite commands in BC BASIC,apart from DEFSPR, begin
with the three letters SPR).

SPRON 0,1

The first 0 is the number of the sprite. The 1 after the comma
may be any non-zero value - it indicates that we want to switch
the sprite ON. If it had been zero, the sprite would have been
switched OFF.

Nothing happened, right? Well, that is because the sprite is
positioned off the screen, where we can’t see it. But first, we
must tell the VIC chip which “sprite block” it is to wuse -
remember that there are 256 sprite blocks in each 16K VIC block
of RAM.

What we are doing is setting the VIC sprite pointers - therefore
we use the SPRPOINT command.

SPRPOINT 0,13

The 0 is the number of the sprite - 13 is the number of the
block. If you defined your sprite in a different sprite block,
say 11, then substitute this instead.

Now we are ready to move the sprite onto the screen. We are
changing its position, so we use the SPRPOS command.

SPRPOS 0,100,100

Again, the 0 is the number of the sprite. The first 100 is the
new x position of the sprite; the second 100 is its y position.

If all has gone well, your sprite should now be visible in all
its glory.

If you don’t 1like its colour, use the SPRINK command, INK
signifying the colour the sprite is drawn in.

SPRINK 0,10

Yet again, the 0 is the sprite number. The 10, which may be any
number in the range 0-15, is the new colour of the sprite -~
light red in this case.

Now try putting another sprite on the screen:
SPRON 5,1:SPRPOINT 5,13:SPRPOS 5,100,150:SPRINK 5,5

Another sprite, this time green and numbered 5, will appear a
little way below the original sprite. Notice that it is getting
its information from the same sprite block as sprite 0, i.e.
block 13 - in fact, all the sprites may access the same block, or
they may all be different.

Part of the power of BC BASIC sprite commands lies in the fact
that many of them have several different formats (e.g. SPRON has
three) so that you may control more that one sprite in each
command .

Time for a few command definitions:

SPRCLR This command turns off all sprites, sets

(sprcL) all positions to 0, clears the X and Y expand
registers, and sets all the sprite INK and
multi-colour colours to the current INK
colour. THIS COMMAND SHOULD BE USED AT THE
BEGINNING OF EVERY SPRITE PROGRAM to ensure
that no spurious sprites are on, and that the
sprites are not expanded etc.

SPRON n,e This command is used to enable or disable
or SPRON n TO n2,e sprites. The first format you have already
or SPRON=n met - if e is non-zero, sprite n is enabled;

if e is zero, sprite n is disabled.
The second format does the same, but for all
sprites from n to n2 inclusive. It does not

SPRINK n[TO n2],col

SPRPOINT n[TO n2],p

(sprpol)

matter which way round n and n2 are.

e.g. SPRON 3 TO 6,1 switches on sprites
3,4,5 and 6.
The third format is used to set the entire
‘sprite enable register’ in VIC in one go, to
n which is in the range 0-255.

e.g. (76543210)

SPRON=%00111011

This switches on sprites 0,1,3,4 and 5, and
switches off sprites 2,6 and 7.

This command is used to set the colour of a
sprite or sprites. If “TO. n2 . -is  net
included, then only sprite n is affected; if
‘TO n2”° is specified, then the colours of all
sprites n to n2 inclusive will be set to
Taoli.

Similar to SPRINK, this command sets the
sprite pointers of one or more sprites to p,
which must be between 0 and 255. If TOma
is not included in the command, then only the
pointer of sprite n is affected; if "TO n2~
is specified, all sprites from n to n2
inclusive will have their pointers set to p.

SPREXP n[TO n2],xe,ye

or SPREXP=xe,ye

(spreX)

SPRX

n,xp

This command is similar to SPRON. It sets
the x and y expansions of one or more
sprites. The first format may specify either
one sprite (n) or several sprites (n TO n2
inclusive). If xe is non-zero, the specified
sprite or sprites are doubled in width in the
x direction; if xe is zero, they are set to
normal width. The same applies to ye for the
height of the sprites.

The second format sets the X and Y expand
registers in the VIC chip in one go, like
SPRON=. xe and ye must be between 0 and 255 -
in binary, their bit patterns set which
sprites will be normal (a 0 bit) and which
will be expanded (a 1 bit).

This command sets the x-position of sprite n
to) Xp. It performs two special functions
which are normally very difficult in standard
BASIC. Firstly, it deals with the nineth bit
of the x-position automatically - so you can
move right the way across the screen without
the hassle involved with attending to the
register for the top bit of the sprite
positions. Secondly, it tells the difference
between PAL and NTSC versions of the
Commodore 64; if xp goes negative, the

computer decides whether the sprite should
move to position 503 (PAL computers) or 511
(NTSC computers). xp should always be
between -504 and +511. You will only need to
use negative values if you are using double-
width sprites.

SPRY n,yp This command sets the y position of sprite n
to yp, which is in the range 0-255. A y=-
position of 0 is off the top of the sacreen;
b i you increase yp, the sprite moves

downwards.

SPRPOS n,xp,yp A combination of the SPRX and SPRY commands,
this command is a convenient way of setting
the position of a sprite in one go. The same
rules apply to xp and yp as if you were using
the SPRX and SPRY commands separately.

A more efficient way of defining sprites

Using 21 DEFSPR commands in a program to define one sprite i
both wasteful of memory and time-consuming. A better way is to
use READ and DATA statements, as follows:

10 FOR K=0 TO 20:READ S$:DEFSPR 13,K,VAL("%"+S$) :NEXTK ?

Continued....

9000 DATA 000000000000000000000000
9001 DATA 000000000000000000000000

6020 DATA 000000000000000000000000

You can create lines 9000-9020 by entering line 9000, and then
moving up the cursor, incrementing the line number and then
pressing RETURN until all 21 lines are present. Then LIST9000-
9020 and you can move the cursor up and define your sprite by
typing a 1 wherever you want the screen to be illuminated.

When you have finalised your sprite, you can compact the program
further (to make it easier for anyone else typing it in) by
converting the binary number in each data statement to either
decimal or hex, and modifing line 10 slightly.

BC BASIC Sprite functions

BC BASIC gives you several functions to read sprite parameters -
to see what colour a sprite is, whether it is on or off, or what
its position on the screen is, for example. Here are some
function definitions:

SPRDEF (n, rw) This function returns the definition of
(sprdkE) sprite n, row rw (in the range 0-20) as a
number in the range 0 to $FFFFFF (which is
L IBEREROBLBHBLED SN 58 8 HE N
e.qg. 10 FOR R=0 TO 20
20 PRINT BINS(SPRDEF(13,R),24,46,209)
30 NEXT R
This prints the definition of sprite 13 as a
matrix of characters 24 across by 21 down.

SPRINK (n) This function returns the colour of sprite n,
as a number in the range 0-15.

SPRPOINT (n) This function returns the current value of

(sprpol) the sprite pointer for sprite n, in the range
0=255.

SPRX(n) This function returns the X position of

sprite n as a number in the range 0-511.
Remember, for PAL computers, -1=503 and for
NTSC computers, -1=511. You only need to
remember this if you are using expanded
sprites. To see what sort of computer you
have, PRINT PEEK($2A6) - 0 is NTSC, 1 is PAL.

SPRY (n) This function returns the Y position of
sprite n as a number between 0 and 255.

SPRON(n) This function, which is used to read the
or SPRON sprite enable register, has two formats, The
first, where it is followed by (n), returns 0
if sprite n is disabled and =1 if aprite n in
enabled (-1 = logical "TRUE).
The second format merely reads the entiiwe
sprite enable register, and returns a valuws
between 0 and 255, where each bit reprosonts
whether that particular sprite is on or off,

SPREXPX(n) This function is to check if a sprite iIn
or SPREXPX expanded in the x direction or not, The
(spreXx) first format, followed by (n), checks mprite

n, returning 0 if it is unexpanded o1 [
it is expanded in the x direction. The second

format reads the entire sprite x-oxpand
register, returning a value between 0 and
256
SPREXPY (n) As for SPREXPX, but for expansion in the vy
or SPREXPY direction rather than the x direction.

(spreXy)

Sprite Priorities

This is very simple. When two or more sprites are on top of each
other, the sprite with the lowest number will be displayed no
sprite 0 is displayed on top of all the other sprites, sprite |
is on top of sprites 2-7 but beneath sprite 0, and sprite 7 in
always bottom.

A demonstration sprite program

10 SPRCLR:SPRPOINT 0TO7,11:SPRON 0TO7,1:PRINT"[cls]"
20 FOR K=0 TO 7:SPRINK K,K+2:NEXT K

30 FOR K=0 TO 20:READ A$:DEFSPR 11,K,VAL("%"+A$):NEXTK
40 FOR S=1 TO 20

50 FOR K=0 TO 7:SPRPOS K,30+S*K,120+K*2:NEXT K

60 NEXT S

70 FOR K=0 TO 7

80 FOR Y=SPRY(K)TOOSTEP-2:SPRY K,Y:NEXT Y

90 NEXT K:SPRON=0:END

100 DATA 000000000000000000000000

101 DATA 000000000000000000000000

. define your own sprite in lines 100-120 inclusive
120 DATA 000000000000000000000000
Lines 10-30 initialise the sprites; lines 40-60 make them spread
out across the screen; lines 70-90 make them move up and off the

screen, one by one. The 21 lines from 100-120 inclusive are for
you to define your sprite.

Try adding 15 SPREXP 0 TO 7,1,1 to enlarge the
sprites.

A VERY IMPORTANT POINT

By now, you will be writing sprite programs worth saving.
However, since having sprites on the screen slows down the
processor chip slightly, this creates errors with important
cassette and disk timings. Therefore, ALWAYS switch off all
sprites (the best way is to type SPRCLR) before any cassette or
disk load or save operations, or ?LOAD ERRORS may result.

23

Chapter 4

BASIC INPUT/OUTPUT

To enable you to make full use of the sprite and user-defined
graphics facilities you have already learnt, this chapter deals
with reading the keyboard and joysticks. More advanced 1/0,
namely using the wuser-port, defining the functions keys and
scanning individual keys on the keyboard directly, are covered in
chapter 11.

Firstly, here are two functions to read which key is being
pressed.

KEY This function returns the ASCII code of the

(kE) key which is currently being pressed, or 0 if
no key is pressed. Unlike GET, KEY repeatn
if a key is held.

KEYS$ This function returns the key which in

(KES$) currently being pressed as a string, or a
null string ("") if no key is pressed. Like
KEY, this function also repeats if the key is
held.

e.g. this could be an extract from a game
program:

50 IF KEY$="A" THEN X=X-1:REM MOVE LEFT
60 IF KEY$="S" THEN X=X+1:REM MOVE RIGHT
70 IF KEY=13 THEN GOTO 200:REM FIRE!

In line 70, 13 is the ASCII code for RETURN =
i.e. the return key is the fire button.

Now for three powerful functions to read the joysticks.

JOY (n) This function is used to read joystick n,
or JOY(n,b) where n is 1 or 2. If the format JOY(n) is
(jO) used, the number returned is in the range 0~

31. 1In binary, bit 0 (the right-hand one) is
UP, bit 1 is DOWN, bit 2 is LEFT, bit 3 is
RIGHT and bit 4 is FIRE. Each bit is 0 if the
switch is open, 1 if it is closed.

The second format allows you to specify which
bit to read - b is in the range 0-4, asn
above. 0 is returned if the switch is open;
-1 (logical ‘TRUE’) if it is closed.

e.g. JOY(2,4) returns 0 if the fire button of
joystick 2 is not pressed, -1 if it is.

JOYX(n) This function returns the x-movement in

(jox) Jjoystick'n;, i.e.  =1if it is pulled to . the
left,, 0 if . it 4is centred and 1 if it ds
pulled to the right. It is independent of
movement in the y (up and down) direction.

sy

JOYY (n) As for JOYX, but returns movement in the y-

(joy) direction:; i.e. =1 A4if the stick is pulled
up, 0 if it is centred in the y-direction,
and 1 if it is pulled down.

Together, the JOYX and JOYY functions make
movement a cinch!

e.g. 10 X=100:Y=100
20 SPRPOS 0,X,Y {
30 X=X+JOYX(1l):Y=Y+JOYY(1):GOTO 20

This is all the code that is required to move
a sprite around the screen under Jjoystick
control -~ even diagonally.

Chapter 5

A FIRST LOOK AT STRUCTURED PROGRAMMING

Take the following line of BASIC:
100 GOSUB 13500

On its own, it is meaningless - it gives no indication of what
function the subroutine at 13500 performs. The following in
little better:

100 GOSUB 13500:REM DISPLAY MENU

since it 1is clumsy, and requires you to remember when writing
your program that the ’‘Display Menu  subroutine is at line 13500,

Another problem is that the subroutine at line 13500 may unse
variables which are already in use, disrupting their contents by
the time it returns. Clearly this is completely unacceptable!

BC BASIC overcomes these limitations by means of neat programming
structures known as procedures. They are being introduced here
as you will find them useful in the next chapter, which dealns
with high-resolution graphics.

In their simplest form, procedures are named subroutines. For
example:

13500 DEFPROC MENU
13510 REM DISPLAY MENU, HERE WE GO...

. program code
13800 ENDPROC

In line 13500, DEPROC (DEFine PROCedure) indicates the start ol
the procedure called "MENU’. In line 13800, ENDPROC is the
equivalent of RETURN - it returns to whatever line the procedure
was called from. It is called by:

100 PROC MENU
(pro)

A few notes at this stage:

There are few limitations to procedure names. They must not
contain graphic symbols, double-quotes (shift and 2) or left-hand
brackets. Apart from this, they may be of any length, and may
even contain keywords like PRINT and LET. DEF and PROC must be
the first two words on a line (although spaces may be inserted
between DEF and PROC). Spaces between PROC and the beginning of
the procedure name are irrelevant, but spaces actually in the
procedure name must match exactly. Perhaps most importantly, YOU

CANNOT EXECUTE A ‘DEPROC” - it will give a ?SYNTAX ERROR if you
try, Eto. Instead, the words DEF and PROC are searched for when
you call a procedure. It is good practice to keep all procedures
at the end of a program.

The next important facility of procedures is the ability to use
local variables - in effect, variables which can only be ’seen’
inside the procedure itself. Upon returning, the values of local
variables are restored. The following command is used to define
local variables:

LOCAL variable(s)
(loC)

If there is more than one variable, separate them by commas.

This command remembers the old contents of variables, so they {
will be restored by the ENDPROC command. Variables may be

normal, integer, string, or elements of an array.

e.g. LOCAL A,F%,C$,A(3),B%(9,7),S%$(2)

It should be obvious that LOCAL may only be used within a
procedure.

An even more powerful feature of BC BASIC procedures is that
parameters, or numbers to be used, may be passed in brackets.
For example:

13500 DEFPROC MENU (A,B)

If this is called by PROC MENU (3,5) then A is set to 3 and B set
4 to 5 - PLUS A and B are automatically defined as LOCAL, so that
if they were being used before, their old values will be
restored. This facility is extremely powerful - even strings may
be passed to procedures.

A final point: like GOSUBs, PROCedures may call other procedures,
still with all the LOCAL variable facilities.

Here is an example of using procedures:

Type: 10 DEFPROC TEST(A$,B,C):LOCAL D
20 PRINT"A$=";A$:PRINT"B=";B:PRINT"C=";C
30 D=B+C:PRINT"D=B+C=";D:ENDPROC

Now type: A$="ABC":B=39:C=87:D=22
PROC TEST("HELLO",3,7)
PRINT AS$,B,C,D

You should find that A$,B,C, and D have kept their original
values, even though they are used in the procedure.

IMPORTANT POINT: If you wish to make sure that your procedures
are completely self-contained, then do not use FOR...NEXT loops
inside them. The reasons for this are complex, but are to do

with the fact that 64 BASIC keeps a separate record of FOR...NEXT
loops (as well as the value of the variable) and if you terminate
such a loop within a procedure, BASIC thinks the same loop is
terminated outside the procedure (if one was in use with the same l

|

variable).

You should try to get into the habit of breaking your program \ :
into chunks, and then defining them as PROCedures - it makes
program writing and debugging much faster and easier.

Chapter 6

HIGH-RESOLUTION GRAPHICS

We now come to a very important aspect of the Commodore 64 - bit-
mapped graphics, or high-resolution graphics, or just hi-res for
short. It is possibly one of the main reasons why you bought the
BC BASIC cartridge.

Bit-mapping, a very popular computer graphics technique, allows
you to control each individual DOT on the screen, independant of
the others; to draw lines of any length and direction; and to
create very detailed pictures, graphs and diagrams.

BC BASIC has two hi-res graphics modes, MODE 3 and 4, which will
be explained in detail shortly - however, briefly, MODE 3 allows
a resolution of 200 dots down the screen by 320 dots across it,
with a foreground and background colour selectable for each 8 by
8 dot square; and MODE 4 allows a resolution of 200 dots down by
160 dots across, but each square may contain dots in three diffe-
rent colours, plus the background, which is the same for the
whole screen.

You have two hi-res screens to use, numbered 0 and 1, which are
totally independent of each other - so you may be displaying one
of them while you are preparing the other one to be displayed.
You may also put text on the hi-res screens using the HPRINT
command, and this text may be single or double-width. In MODE 3,
HPRINT has the advantage over normal text mode that each
character on the screen may have its own PAPER colour as well as
its own INK colour, but it is slightly slower that PRINT
(nevertheless, it is fast enough for most applications).

The two hi-res screens are separate from the text screen - you
must display either one or the other. Therefore, you cannot see
what you are doing in hi-res mode, but programs may output to the
screen using HPRINT, e.g. for labelling graphs.

Telling the computer to use hi-res mode

The command MODE is used to change graphics modes. For the
moment we shall concentrate on MODE 3, the STANDARD BIT-MAP MODE,
but we shall look at MODE 4 later.

To switch into hi-res mode, type the following:

MODE 3,0:CLG
(mO)

The 3 tells the computer which mode you wish to enter - in this
case, standard hi-res. The 0 after the comma is the number of the

screen you wish to use, which is 0 or 1. The CLG for CLear
Graphics, clears the screen of anything which was on it
beforehand. You don’t have to use a CLG after a MODE command =

we just want a clear screen so we can start experimenting.

